
Perl Gtk3 Programming

Kevin and Joshua Philp

Version Date: 05/09/12

Table Of Contents

1. Introduction.....	1
1.a References.....	2
1.b Interpreting the C reference material.....	3
1.c Getting ready.....	3
2. The Basics.....	3
2.a The object hierarchy.....	3
2.b Events, Signals and Callbacks.....	3
3. Windows.....	4
3.a A simple window.....	4
3.b A better Window.....	7
4. Button Basics.....	8
4.a Adding buttons.....	8
4.b Adding button functions.....	11
5. Labels.....	13
5.a Fun with labels.....	13
5.b Labels with mnemonics.....	17
6. More Buttons.....	17
6.a Check box.....	17
6.b Radio Box.....	20
6.c Spinbox.....	22
6.d Toggle Button.....	24
6.e Scale Button.....	25
7. Layout Containers.....	25
7.a Box.....	25
7.b Grid.....	28
8. Displays.....	29
8.a Progress bar.....	29
8.b Spinner.....	30
8.c Switch.....	32
9. Cairo drawing basics.....	32
10. Dealing with user input.....	32

1. Introduction

This tutorial was written jointly by my son and I. The aim was to learn some perl Gtk3 programming and maybe pass on some of what we learnt.

1.a References

Firstly I need to give credit to all the great sources of information that we came across while putting this together:

The Gtk2 Perl Study Guide
Dirk van der Walt.
<http://gtk2-perl.sourceforge.net/doc/gtk2-perl-study-guide/>
Our main inspiration.

Introduction to GUI Programming with Gtk2-Perl
Ross McFarland
<http://gtk2-perl.sourceforge.net/doc/intro/>

GTK+ 3 Reference Manual
<http://developer.gnome.org/gtk3/stable/>
Although its all in C, this is the basis of Gtk3 and is therefore the canonical source. Most of the bindings to Perl are deliberately logical and it soon becomes easy to interpret.

Un-official Seed Documentation
<http://www.roojs.org/seed/gir-1.2-gtk-3.0/seed/Gtk.html>
The closest I have seen to a detailed Perl Gtk3 documentation.

Gtk2-Perl: An Introduction: Or How I Learned to Stop Worrying and Make Things Pretty
Scott Paul Robertson
<http://media.scottr.org/presentations/gtk2-perl.pdf>

Programming Graphical Applications with Gtk2-Perl, Part 1
<http://www.drdobbs.com/web-development/programming-graphical-applications-with/184416060>

Programming Graphical Applications with Gtk2-Perl, Part 2
<http://www.drdobbs.com/web-development/programming-graphical-applications-with/184416069>

The Python Gtk3 Tutorial
<http://python-gtk-3-tutorial.readthedocs.org/en/latest/index.html>
Several examples are based on Python programs from this tutorial.

1.b Interpreting The C Reference Material

One day you are probably going to have to read the C reference material. Hopefully these notes will save you from that fate for as long as possible but we all end up there eventually. So some basic tips on how to read them:

Enumerations: Either use the full name or the registered nickname in lower case. So GTK_POS_LEFT in a perl program can be either 'GTK_POS_LEFT' or simply 'left'.

1.c Getting Ready

We need to find out which version of perl you are actually running. The version below comes from my installation of Debian Wheezy (testing) on an AMD64 architecture.

```
# perl -v
This is perl 5, version 14, subversion 2 (v5.14.2) built for x86_64-linux-gnu-thread-multi
(with 61 registered patches, see perl -V for more detail)
```

We need to check we have the gtk3 perl library installed.

```
aptitude search libgtk3-perl
i libgtk3-perl                - Perl bindings for the GTK+ graphical user interface
library
```

If it is not installed then install with

```
# aptitude install libgtk3-perl
```

You will need to be root to install a new package.

2. The Basics

2.a The Object Hierarchy

2.b Events, Signals And Callbacks

A Gtk3 program is event driven. What this means is that, unlike a conventional computer program that starts at the beginning and works its way to the end, a Gtk3 program will sit in a continuous loop waiting for user input and acting upon that input. This process can be divided into three basic components:

Event	The user clicks a button, moves the mouse, presses a key or drags an icon. Just about anything the user does will generate an event. An event is passed to the relevant widget – maybe the button you clicked. The important point is that the event causes the widget to generate signal if that event is allowed by the widgets event mask. Widgets have default event masks and for most basic uses you will not have to adjust the event mask.
-------	--

Signal	<p>We need our signal to trigger some code. So we connect signals to a subroutine. This can be either an anonymous subroutine or closure</p> <pre>\$window->signal_connect (delete_event => sub { Gtk3->main_quit });</pre> <p>or a named subroutine.</p> <pre>\$button2->signal_connect (clicked => \&quit_function, \$userdata);</pre> <p>When connecting to an external subroutine or a closure two basic pieces of information are passed as arguments: the first is the widget sending the signal, the second is the event that triggered the signal. You can also add an extra argument of user data if you wish. The use of a closure allows you to use other variables in the current scope. The use of an external subroutine allows you to send different signals to the same routine. So there are advantages to both and we will use both in the tutorial.</p>
Callbacks	<p>The subroutine that is triggered by the signal is called a callback. There is one important point about callbacks and Perl. In Gtk3 a callback that returns false will propagate the signal on to, potentially, other signal handlers. If the callback returns true the signal will not propagate and that could mean that some callbacks are not getting called. The problem with Perl is that by default it will return the value of the last calculation – that maybe not what we want. So to ensure compliant behaviour it is good practice to end your callback subroutines with an explicit return value. In most cases this will be return FALSE.</p>

3. Windows

3.a A Simple Window

We will make a simple window that we can maximise, minimise and drag around the screen.




Figure 1: A Basic Window

```
#!/usr/bin/perl

use strict;
use warnings;
use diagnostics;
use feature ':5.14';
use Gtk3 '-init';

my $window = Gtk3::Window->new('toplevel');
$window->show_all;
Gtk3->main;
```

Once you have typed the above into a file, save as 1-A-basic-window.pl, set the execute permission and run.

```
$ chmod 700 1-A-basic-window.pl
$ ./1-A-basic-window.pl
^C
$
```

Let's have a look at what we have, how it works and then see what it does.

The first line tells the system where the perl compiler is. If you get an error saying:

```
bad interpreter: No such file or directory
```

then check you are pointing to the right place and change the line to point to where your Perl interpreter actually is.

```
# which perl
/usr/bin/perl
```

```
use strict;  
use warnings;
```

The next two lines are highly recommended in all your Perl programs and prevent some bad Perl practices as well as issuing warnings for others.

```
use diagnostics;
```

This is optional, but if you have an error it will give you a more detailed report on the problem. It might help you find

```
use feature ':5.14';
```

This turns on some of the more recent language features such as using

say “Hello”;

in place of

print “Hello\n”;

```
use Gtk3 '-init';
```

This is where we initialise the Gtk3 toolkit. Every script that uses Gtk3 will need this line. A few caveats on using this line:

Gtk3 needs to be initialised before its used. You can declare Gtk3 in the use statement and initialise it later but it is safer to do the two together.

If your program relies several modules then Gtk3 needs to be initialised in the main program, you shouldn't initialise it in every module.

```
my $window = Gtk3::Window->new('toplevel');
```

In this line we initialise a new variable (\$window) and we set it up as a new Gtk3 window. We have a choice of window types, in this case we choose a 'toplevel' window, other types include 'popup' for information and quick dialogue boxes.

```
$window->show_all;
```

We need to tell Gtk3 that we want to show the window on the screen. This statement shows the window and all its children. At the moment we have no children so it just shows the window.

```
Gtk3->main;
```

Finally this causes Gtk3 to wait for user input. But as we have nothing set up to catch that input the window just sits there doing nothing and with little functionality. The window can be dragged around the screen, the maximise and minimise buttons work and if you click the close button the window disappears. However the program as not quitted properly and you still need to press Ctrl-C from the terminal to kill the program and get the command prompt back.

When an event occurs in a Gtk3 application that event generates a signal. With the program as it stands most of those signals are just lost. To add functionality we tell Gtk3 to look out for certain signals and pass them to a function we set up.

The close button on the top of the window generates a “delete_event” signal. We need to look out for that signal and when its triggered we need to shut the program down properly by telling Gtk3 to quit.

To fix the problem we need to add one line.

```
#!/usr/bin/perl

use strict;
use warnings;
use diagnostics;

use Gtk3 '-init';

my $window = Gtk3::Window->new('toplevel');

$window->signal_connect (delete_event => sub { Gtk3->main_quit });

$window->show_all;
Gtk3->main;
```

Now when you click the close button on the window the program shuts down properly.

We could have set up a separate function and ran that when the event was fired. However in this case the function was very short and we could use an anonymous (as opposed to a named) subroutine embedded into the statement.

3.b A Better Window

In this section we will add our own title, centre the window in the screen and set the default size to something a little more sensible.

```
#!/usr/bin/perl

use strict;
use warnings;
use diagnostics;

use Gtk3 '-init';

my $window = Gtk3::Window->new('toplevel');
$window->set_title("My Title");
$window->set_position("center");
$window->set_default_size(400, 300);
$window->set_border_width(20);

$window->signal_connect (delete_event => sub { Gtk3->main_quit });
```

```
$window->show_all;  
Gtk3->main;
```

```
$window->set_title("My Title");
```

The first new line simply replaces the text in the title bar with some text we supply.

```
$window->set_position("center");
```

we cause `set_position` to change where the window will be placed. In this case we have moved it to the centre of the screen. Other options include:

<code>none</code>	Don't influence where the window is created.
<code>center</code>	Create the window in the centre of the screen.
<code>mouse</code>	Create the window under the current mouse pointer.

```
$window->set_default_size(400, 300);
```

This sets the default width and height of the window and is measured in pixels.

```
$window->set_border_width(20)
```

Adds a thicker border to the window.

4. Button Basics

4.a Adding Buttons

In Gtk3 terms a child is an element that is created inside another – the parent. Hence your Gtk3 program will look like a tree structure with your window object containing its children who may then contain their children.

A Gtk3 window has a major restriction, it can only have one child. So you can't add multiple buttons or any other widgets directly to a window. To handle this we first add a container widget and this will hold our buttons and any other widgets, maybe even different containers..

A container widget is a box for other widgets to fit into. Its purpose is to organise the layout of your window. You tell it how you want your widgets laid out. In the basic box container you tell the container whether to stack the widgets on top of each other (vertical) or line them up next to each other (horizontal).

In this section we will had two buttons to the window and connect both to a quit function.



Figure 2: A Better Window

```
#!/usr/bin/perl

use strict;
use warnings;
use diagnostics;
use feature ':5.14';
use Gtk3 '-init';

use Glib qw/TRUE FALSE/;

sub quit_function {
    say "Exiting Gtk3";
    Gtk3->main_quit;
    return FALSE;
}

my $window = Gtk3::Window->new('toplevel');
$window->set_title("My Title");
$window->set_position("mouse");
$window->set_default_size(400, 50);
$window->set_border_width(20);
$window->signal_connect(delete_event => \&quit_function);

my $button1 = Gtk3::Button->new("Quit");
$button1->signal_connect(clicked => \&quit_function);

my $button2 = Gtk3::Button->new("Another Quit");
$button2->signal_connect(clicked => \&quit_function);

my $hbox = Gtk3::Box->new("horizontal", 5);
$hbox->pack_start($button1, TRUE, TRUE, 0);
$hbox->pack_start($button2, TRUE, TRUE, 0);

$hbox->set_homogeneous(TRUE);

$window->add($hbox);

$window->show_all;
Gtk3->main;
```

In this program we have added a few needs lines.

```
use Glib qw/TRUE FALSE/;
```

Perl has no predefined values for true and false, it uses 1 and 0. To make your programs more readable it is good practice to import the predefined values already defined in the Glib library. This line imports the Glib values for TRUE and FALSE so they can be used.

```
sub quit_function {
    say "Exiting Gtk3";
    Gtk3->main_quit;
    return FALSE;
}
```

The next new item is a small function. It will take over the place of the anonymous function in the previous program that we used to close the program down. The function will print a short closing message to the terminal, quit Gtk3 and destroy the window. Any button we want to use to close the program will now point to this function.

```
$window->signal_connect (delete_event => \&quit_function);
```

The next new line is where we connect the delete event to our new quit_function. Note that the signal connect function requires a reference to a function and not the function name itself. To make a reference in Perl you prefix the function name with \&.

```
my $button1 = Gtk3::Button->new("Quit");
$button1->signal_connect (clicked => \&quit_function);

my $button2 = Gtk3::Button->new("Another Quit");
$button2->signal_connect (clicked => \&quit_function);
```

In this section of code we set up two buttons and connect the clicked event for each of them to our new quit_function.

```
my $hbox = Gtk3::Box->new("horizontal", 5);
$hbox->pack_start($button1, TRUE, TRUE, 0);
$hbox->pack_start($button2, TRUE, TRUE, 0);

$hbox->set_homogeneous (TRUE);

$window->add($hbox);
```

Finally we define a new box (our container) and pass it two parameters. The first tells Gtk3 that we want to stack items in the box horizontally and the second parameter is the spacing between the widgets in pixels.

Now we have a box we start adding things to it. To do this we use the pack_start function which packs widgets into the container starting at the left hand edge and adding more to the right.

The pack_start function has several extra parameters which are:

```
$hbox->pack_start($button2, expand, fill, padding);
```

Expand : TRUE – expand the container to make space for the button.

Fill : TRUE – the extra space created by the expand option above is allocated to the widgets and not just filled with padding.

Padding : An extra number of pixels to add between widgets to create some space.

```
$hbox->set_homogeneous (TRUE);
```

The set_homogeneous function tells the box to allocate the same amount of space to each widget, so we get two buttons the same width;

```
$window->add($hbox);
```

Finally we had our new container to the main window.

4.b **Adding Button Functions**

Now we are going to get our buttons to do something more interesting. We will get them to change the text on a label.



Figure 3: Adding Button Functions

```
#!/usr/bin/perl

use strict;
use warnings;
use diagnostics;
use feature ':5.14';
use Gtk3 '-init';

use Glib qw/TRUE FALSE/;

#### DECLARATIONS

my $window = Gtk3::Window->new('toplevel');
$window->set_title("My Title");
$window->set_position("mouse");
$window->set_default_size(400, 50);
$window->set_border_width(20);
$window->signal_connect (delete_event => \&quit_function);
```

```

my $button1 = Gtk3::Button->new("Quit");
$button1->signal_connect (clicked => \&quit_function);

my $button2 = Gtk3::Button->new("Say Hello");
$button2->signal_connect (clicked => \&say_something, "Hello");

my $button3 = Gtk3::Button->new("Say Goodbye");
$button3->signal_connect (clicked => \&say_something, "Goodbye");

my $hbox = Gtk3::Box->new("horizontal", 5);
$hbox->set_homogeneous (TRUE);
$hbox->pack_start($button1, TRUE, TRUE, 0);
$hbox->pack_start($button2, TRUE, TRUE, 0);
$hbox->pack_start($button3, TRUE, TRUE, 0);

my $vbox = Gtk3::Box->new("vertical", 5);
$vbox->add($hbox);

my $label = Gtk3::Label->new("Am I connected?");
$vbox->add($label);

$window->add($vbox);
$window->show_all;

#### FUNCTIONS

sub quit_function {
    say "Exiting Gtk3";
    Gtk3->main_quit;
    return FALSE;
}

sub say_something {
    my ($button, $userdata) = @_;
    $label->set_label( $userdata );
    return TRUE;
}

Gtk3->main;

```

```

my $button2 = Gtk3::Button->new("Say Hello");
$button2->signal_connect (clicked => \&say_something, "Hello");

my $button3 = Gtk3::Button->new("Say Goodbye");
$button3->signal_connect (clicked => \&say_something, "Goodbye");

```

The button functions have changed slightly. We now declare two buttons, both point to the same function but they also contain a second parameter which will be passed to the function when its activated. So if you press button2 you will activate the say_something function and pass it the string “Hello”. If you press button3 you will activate the same function but this time you will pass it the

string "Goodbye".

```
my $vbox = Gtk3::Box->new("vertical", 5);  
$vbox->add($hbox);  
  
my $label = Gtk3::Label->new("Am I connected?");  
$vbox->add($label);  
  
$window->add($vbox);
```

The buttons are packed into a horizontal box as before but that box is not added directly to the window. Instead we create another box, a vertical box, and we add the horizontal button box to this, followed by a new label with the text "Am I connected". The effect is to have a row of button with a label underneath them.

Finally we add the vbox (which has the hbox and label) to the window. Our object hierarchy now looks like.




Figure 4: Object Hierachy

5. Labels

5.a Fun With Labels

A demo of how you can style labels using Pango markup and other effects.




Figure 5: Fun With Labels

```
#!/usr/bin/perl

use strict;
use warnings;
use diagnostics;
use feature ':5.14';
use Gtk3 '-init';

use Glib qw/TRUE FALSE/;

#### DECLARATIONS

my $window = Gtk3::Window->new('toplevel');
$window->set_title("My Title");
$window->set_position("mouse");
$window->set_default_size(400, 50);
$window->set_border_width(20);
$window->signal_connect (delete_event => \&quit_function);

my $button1 = Gtk3::Button->new("Twist");
$button1->signal_connect (clicked => \&say_something, "twist");

my $button2 = Gtk3::Button->new("red");
$button2->signal_connect (clicked => \&say_something, "red");

my $button3 = Gtk3::Button->new("weird");
```

```

$button3->signal_connect(clicked => \&say_something, "weird");

my $hbox = Gtk3::Box->new("horizontal", 5);
$hbox->set_homogeneous(TRUE);
$hbox->pack_start($button1, TRUE, TRUE, 0);
$hbox->pack_start($button2, TRUE, TRUE, 0);
$hbox->pack_start($button3, TRUE, TRUE, 0);

my $vbox = Gtk3::Box->new("vertical", 5);
$vbox->add($hbox);

my $label = Gtk3::Label->new("Am I connected?");
$vbox->add($label);

$window->add($vbox);
$window->show_all;

#### FUNCTIONS

sub quit_function {
    say "Exiting Gtk3";
    Gtk3->main_quit;
    return FALSE;
}

sub say_something {
    my ($button, $userdata) = @_;
    if ($userdata eq "twist") {
        my $angle = $label->get_angle;
        $label->set_angle($angle - 30);
    } elsif ($userdata eq "red") {
        $label->set_markup('<span foreground="red"><i>Red and connected</i></span>');
    } elsif ($userdata eq "weird") {
        $label->set_markup(
            '<span foreground="cyan" background="yellow" size="x-large"><b>Weird</b></span>');
    }
    return FALSE;
}

Gtk3->main;

```

```

my $button1 = Gtk3::Button->new("Twist");
$button1->signal_connect(clicked => \&say_something, "twist");

my $button2 = Gtk3::Button->new("red");
$button2->signal_connect(clicked => \&say_something, "red");

my $button3 = Gtk3::Button->new("weird");

```



```
$button3->signal_connect(clicked => \&say_something, "weird");
```

We declare three buttons, give them different names and connect each ones clicked event to the same function (say_something). Each button passes a different parameter telling the function what it wants to happen to the label.

```
sub say_something {
    my ($button, $userdata) = @_;
    if ($userdata eq "twist") {
        my $angle = $label->get_angle;
        $label->set_angle( $angle - 30 );
    } elsif ($userdata eq "red") {
        $label->set_markup('<span foreground="red"><i>Red and connected</i></span>');
    } elsif ($userdata eq "weird") {
        $label->set_markup(
            '<span foreground="cyan" background="yellow" size="x-large"><b>Weird</b></span>');
    }
    return FALSE;
}
```

The rest of the program is the same as before until we get to the say_something function. This time we check the userdata parameter passed and trigger a different label function depending on what we receive.

If the userdata is “twist” we read the label angle, take away 30 degrees and reset the label to the new angle. Each time we press the button the label moves by another 30 degrees.

If the userdata is “red” we change the text using the set_markup function which allows us to add html type styling. If the userdata is weird we change the styling to cyan with a yellow background.

5.b Labels With Mnemonics

6. More Buttons

6.a Check Box

This chapter is heavily based on the Python Gtk3 tutorial.

(<http://python-gtk-3-tutorial.readthedocs.org/en/latest/index.html>).




Figure 6: Check Box Example

In this tutorial we set up a simple text box and toggle a few of the properties. Specifically we will toggle whether the box can be edited or not, whether the letters will be obscured (like a password entry box) and finally we will toggle an icon on and off.

In outline we set up a vertical box and put the entry widget at the top followed by a horizontal box underneath that will contain our check boxes. Each check box is set up its own label and each check box points to the same call back function. The call back function identifies which button called it using the supplied user data string and acts accordingly.

```
#!/usr/bin/perl

#
#   This example borrows very heavily from the
#   The Python GTK+ 3 Tutorial
#   http://python-gtk-3-tutorial.readthedocs.org/en/latest/index.html
#

use strict;
use warnings;
use diagnostics;
use feature ':5.14';
use Gtk3 '-init';
use Glib qw/TRUE FALSE/;

my $window = Gtk3::Window->new('toplevel');
$window->set_title("Basic Check Boxes");
$window->set_position("mouse");
$window->set_default_size(200, 100);
$window->set_border_width(5);
$window->signal_connect (delete_event => sub { Gtk3->main_quit });

my $vbox = Gtk3::Box->new("vertical", 5);
$window->add($vbox);

my $entry = Gtk3::Entry->new;
$entry->set_text("Hello World");
$vbox->pack_start($entry, TRUE, TRUE, 0);

my $hbox = Gtk3::Box->new("horizontal", 6);
```

```

$ vbox->pack_start($ hbox, TRUE, TRUE, 0);

my $check_button1 = Gtk3::CheckButton->new_with_label('Editable');
$check_button1->signal_connect (toggled => \&cb_toggled, "editable");
$check_button1->set_active(TRUE);
$hbox->pack_start($check_button1, TRUE, TRUE, 0);

my $check_button2 = Gtk3::CheckButton->new_with_label("Visible");
$check_button2->signal_connect (toggled => \&cb_toggled, "visible");
$check_button2->set_active(TRUE);
$hbox->pack_start($check_button2, TRUE, TRUE, 0);

my $check_button3 = Gtk3::CheckButton->new_with_label("Icon");
$check_button3->signal_connect (toggled => \&cb_toggled, "icon");
$check_button3->set_active(FALSE);
$hbox->pack_start($check_button3, TRUE, TRUE, 0);

$window->show_all;
Gtk3->main;

sub cb_toggled {
    my ($widget, $data) = @_;
    my $value = $widget->get_active;

    if ($data eq 'visible') {
        $entry->set_visibility($value);
    } elsif ($data eq 'editable') {
        $entry->set_editable($value);
    } elsif ($data eq 'icon') {
        my $stock_id = undef;
        if ($value == TRUE) {
            $stock_id = Gtk3::STOCK_FIND;
        }
        $entry->set_icon_from_stock('secondary', $stock_id);
    }
    return FALSE;
}

```

The first section of the program follows the same outline as previous programs. After that the section below first declares a new checkbox with an attached label that will be displayed alongside.

The checkboxes are each set up with a `signal_connect` statement that connects the “toggled” event with the `cb_toggled` subroutine and also sends the name of the button in the optional extra string.

The `set_active` function defines if the initial state of the button is ticked or not.

Finally the button is packed into the horizontal container set up below the entry box. The other buttons follow a similar pattern.

```

my $check_button1 = Gtk3::CheckButton->new_with_label('Editable');
$check_button1->signal_connect (toggled => \&cb_toggled, "editable");

```

```
$check_button1->set_active(TRUE);  
$hbox->pack_start($check_button1, TRUE, TRUE, 0);
```

The `call_back` function extracts the widget and supplied data and then determines if the widget is ticked or not (`$widget->get_active`). The series of if/else statements then decide what action to take.

If `$data` is 'visible' then the visibility of the text in the entry box is set according to the `$value` i.e. if `$value` is TRUE (ticked) then the `$entry->set_visibility` is also set TRUE. The editable check box follows the same pattern.

If the icon tick box is toggled on then a stock icon is loaded and placed in the entry box. Note the position parameter: primary is at the start of the box, secondary is at the end. If we set the stock icon to undef we simply have no icon.

```
sub cb_toggled {  
    my ($widget, $data) = @_;  
    my $value = $widget->get_active;  
  
    if ($data eq 'visible') {  
        $entry->set_visibility($value);  
    } elsif ($data eq 'editable') {  
        $entry->set_editable($value);  
    } elsif ($data eq 'icon') {  
        my $stock_id = undef;  
        if ($value == TRUE) {  
            $stock_id = Gtk3::STOCK_FIND;  
        }  
        $entry->set_icon_from_stock('secondary', $stock_id);  
    }  
    return FALSE;  
}
```

6.b Radio Box




Figure 7: Radio Box Example

```

#!/usr/bin/perl

use strict;
use warnings;
use diagnostics;
use feature ':5.14';
use Gtk3 '-init';
use Glib qw/TRUE FALSE/;

my $window = Gtk3::Window->new('toplevel');
$window->set_title("Basic Radio Boxes");
$window->set_position("mouse");
$window->set_default_size(200, 100);
$window->set_border_width(5);
$window->signal_connect (delete_event => sub { Gtk3->main_quit });

my $hbox = Gtk3::Box->new("horizontal", 5);
$window->add($hbox);

my $rbutton1 = Gtk3::RadioButton->new_with_label_from_widget(undef, "Button 1");
$hbox->pack_start($rbutton1, TRUE, TRUE, 0);
$rbutton1->signal_connect (toggled => \&rb_toggled, "rb1");

my $rbutton2 = Gtk3::RadioButton->new_with_label_from_widget($rbutton1, "Button 2");
$hbox->pack_start($rbutton2, TRUE, TRUE, 0);
$rbutton2->signal_connect (toggled => \&rb_toggled, "rb2");

```

```

my $rbutton3 = Gtk3::RadioButton->new_with_label_from_widget($rbutton1, "Button 3");
$hbox->pack_start($rbutton3, TRUE, TRUE, 0);
$rbutton3->signal_connect (toggled => \&rb_toggled, "rb3");

$window->show_all;
Gtk3->main;

sub rb_toggled {
    my ($widget, $data) = @_;
    print "Radio button $data is now ";
    if ($widget->get_active == TRUE) {
        say " set ON";
    } else {
        say " set OFF";
    }
    return FALSE;
}

```

6.c Spinbox




Figure 8: Spinbox Example

```

#!/usr/bin/perl

use strict;
use warnings;
use diagnostics;
use feature ':5.14';
use Gtk3 '-init';
use Glib qw/TRUE FALSE/;

my $window = Gtk3::Window->new('toplevel');
$window->set_title("Spinbox Example");
$window->set_position("mouse");
$window->set_default_size(250, 100);
$window->set_border_width(5);
$window->signal_connect (delete_event => sub { Gtk3->main_quit });

my $vbox = Gtk3::Box->new("vertical", 5);

```

```
$window->add($vbox);

my $adjust = Gtk3::Adjustment->new(5, 0, 10, 1, 10, 0);
my $spin = Gtk3::SpinButton->new($adjust, 1, 1);
$vbox->pack_start($spin, FALSE, FALSE, 0);

my $hbox = Gtk3::Box->new("horizontal", 5);
$vbox->pack_start($hbox, FALSE, FALSE, 0);

my $zero_button = Gtk3::Button->new('Zero');
$hbox->pack_start($zero_button, FALSE, FALSE, 0);
$zero_button->signal_connect ( clicked => \&zero );

my $check_accuracy = Gtk3::CheckButton->new_with_label('High accuracy');
$hbox->pack_start($check_accuracy, FALSE, FALSE, 0);
$check_accuracy->signal_connect ( toggled => \&toggle_accuracy );

my $hbox2 = Gtk3::Box->new("horizontal", 5);
$vbox->pack_start($hbox2, FALSE, FALSE, 0);

my $update_button = Gtk3::Button->new('Update');
$hbox2->pack_start($update_button, FALSE, FALSE, 0);
$update_button->signal_connect ( clicked => \&update );

my $label = Gtk3::Label->new("No Value");
$hbox2->pack_start($label, FALSE, FALSE, 0);

$window->show_all;
Gtk3->main;

sub zero {
    my ($widget, $data) = @_ ;
    $spin->set_value(0);
    return FALSE;
}

sub toggle_accuracy{
    my ($widget, $data) = @_ ;
    my $value = $widget->get_active;
    if ( $value == TRUE ) {
        $spin->set_digits(2);
    } else {
        $spin->set_digits(1);
    }
    return FALSE;
}

sub update {
    $label->set_label( $spin->get_value );
    return FALSE;
}
```

6.d Toggle Button




Figure 9: Toggle Button Example

```
#!/usr/bin/perl

use strict;
use warnings;
use diagnostics;
use feature ':5.14';
use Gtk3 '-init';
use Glib qw/TRUE FALSE/;

my $window = Gtk3::Window->new('toplevel');
$window->set_title("Spin and toggle Example");
$window->set_position("mouse");
$window->set_default_size(250, 100);
$window->set_border_width(5);
$window->signal_connect (delete_event => sub { Gtk3->main_quit });

my $vbox = Gtk3::Box->new("vertical", 5);
$window->add($vbox);

my $adjust = Gtk3::Adjustment->new(5, 0, 10, 1, 10, 0);
my $spin = Gtk3::SpinButton->new($adjust, 1, 1);
$vbox->pack_start($spin, FALSE, FALSE, 0);

my $hbox = Gtk3::Box->new("horizontal", 5);
$vbox->pack_start($hbox, FALSE, FALSE, 0);

my $zero_button = Gtk3::Button->new('Zero');
$hbox->pack_start($zero_button, FALSE, FALSE, 0);
$zero_button->signal_connect ( clicked => \&zero );

my $toggle_accuracy = Gtk3::ToggleButton->new_with_label('High accuracy');
$hbox->pack_start($toggle_accuracy, FALSE, FALSE, 0);
$toggle_accuracy->signal_connect ( toggled => \&toggle_accuracy );

my $hbox2 = Gtk3::Box->new("horizontal", 5);
```



```

$ vbox->pack_start($ hbox2, FALSE, FALSE, 0);

my $update_button = Gtk3::Button->new('Update');
$hbox->pack_start($update_button, FALSE, FALSE, 0);
$update_button->signal_connect ( clicked => \&update );

my $label = Gtk3::Label->new("No Value");
$hbox->pack_start($label, FALSE, FALSE, 0);

$window->show_all;
Gtk3->main;

sub zero {
    my ($widget, $data) = @_;
    $spin->set_value(0);
    return FALSE;
}

sub toggle_accuracy{
    my ($widget, $data) = @_;
    my $value = $widget->get_active;
    if ( $value == TRUE ) {
        $spin->set_digits(2);
    } else {
        $spin->set_digits(1);
    }
    return FALSE;
}

sub update {
    $label->set_label( $spin->get_value );
    return FALSE;
}

```

6.e Scale Button

7. Layout Containers





7.a Box

We have used the box extensively over the last few chapters. In the table below are some examples of the different types of layout possible with combinations of the box property homogeneous and setting the fill and expand properties of the objects in the container.

Box orientation is set vertical, spacing between buttons is set to 3.


```
my $vbox = Gtk3::Box->new("vertical", 3);
```



<p>Box orientation is set horizontal, spacing is set to zero. Homogeneous set to TRUE.</p> <pre>my \$hbox = Gtk3::Box->new("horizontal", 0);</pre> <p>Buttons added with expand and fill set to TRUE</p> <pre>\$hbox->pack_start(\$button, TRUE, TRUE, 0);</pre>	
<p>Box orientation is set horizontal, spacing is set to zero. Homogeneous set to TRUE.</p> <pre>my \$hbox = Gtk3::Box->new("horizontal", 0);</pre> <p>Buttons added with expand and fill set to FALSE</p> <pre>\$hbox->pack_start(\$button, FALSE, FALSE, 0);</pre>	
<p>Box orientation is set horizontal, spacing is set to zero. Homogeneous set to FALSE.</p> <pre>my \$hbox = Gtk3::Box->new("horizontal", 0);</pre> <p>First three buttons added with expand and fill set to FALSE</p> <pre>\$hbox->pack_start(\$button, FALSE, FALSE, 0);</pre> <p>Last button added with expand and fill set to TRUE.</p> <p>Expand tells the button to grab any extra free space. If fill is TRUE the extra space is filled with the button, if fill is false the extra space is filled with padding.</p>	
<p>Box orientation is set horizontal, spacing is set to zero. Homogeneous set to FALSE.</p> <pre>my \$hbox = Gtk3::Box->new("horizontal", 0);</pre> <p>First three buttons added with expand and fill set to FALSE</p> <pre>\$hbox->pack_start(\$button, FALSE, FALSE, 0);</pre> <p>Last button added with expand and fill set to TRUE.</p>	

7.b Grid

Grid has replaced table in Gtk2 and has a very similar functionality.



```
#!/usr/bin/perl

use strict;
use warnings;
use diagnostics;
use feature ':5.14';
use Gtk3 '-init';
use Glib qw/TRUE FALSE/;

my $window = Gtk3::Window->new('toplevel');
$window->set_title("Grid Example");
$window->set_position("mouse");
$window->set_default_size(250, 100);
$window->set_border_width(5);
$window->signal_connect (delete_event => sub { Gtk3->main_quit });

my $grid = Gtk3::Grid->new;
$window->add($grid);

my $toggle1 = Gtk3::ToggleButton->new_with_label('One');
my $toggle2 = Gtk3::ToggleButton->new_with_label('Two');
my $toggle3 = Gtk3::ToggleButton->new_with_label('Three');
my $toggle4 = Gtk3::ToggleButton->new_with_label('Four');
my $toggle5 = Gtk3::ToggleButton->new_with_label('Five');
my $toggle6 = Gtk3::ToggleButton->new_with_label('Six');

#attach (Widget child, gint32 left, gint32 top, gint32 width, gint32 height) : none
#void gtk_grid_attach_next_to (GtkGrid *grid, GtkWidget *child, GtkWidget *sibling,
GtkPositionType side, gint width, gint height);


$grid->add($toggle1);
$grid->attach($toggle2, 1, 0, 2, 1);
$grid->attach_next_to($toggle3, $toggle1, "bottom", 1, 2);
$grid->attach_next_to($toggle4, $toggle3, "right", 2, 1);
$grid->attach($toggle5, 1, 2, 1, 1);
$grid->attach_next_to($toggle6, $toggle5, "right", 1, 1);

$window->show_all;
```

```
Gtk3->main;
```

8. Displays

8.a Progress Bar



```
#!/usr/bin/perl

use strict;
use warnings;
use diagnostics;
use feature ':5.14';
use Gtk3 '-init';
use Glib qw/TRUE FALSE/;

my $window = Gtk3::Window->new('toplevel');
$window->set_title("Progress Bar");
$window->set_position("mouse");
$window->set_default_size(250, 100);
$window->set_border_width(5);
$window->signal_connect(delete_event => sub { Gtk3->main_quit });

my $vbox = Gtk3::Box->new("vertical", 5);
$window->add($vbox);

my $progress = Gtk3::ProgressBar->new;
$progress->set_orientation("horizontal");
$progress->set_inverted(FALSE);
$progress->set_text(undef);
$progress->set_show_text(FALSE);
$vbox->add($progress);

my $hbox = Gtk3::Box->new("horizontal", 2);
$vbox->add($hbox);

my $toggle1 = Gtk3::ToggleButton->new_with_label('Text');
my $toggle2 = Gtk3::ToggleButton->new_with_label('Percent');
my $toggle3 = Gtk3::ToggleButton->new_with_label('Invert');
$hbox->pack_start($toggle1, TRUE, TRUE, 0);
$hbox->pack_start($toggle2, TRUE, TRUE, 0);
```

```

$ hbox->pack_start($toggle3, TRUE, TRUE, 0);

$toggle1->signal_connect ( toggled => \&showtext );
$toggle2->signal_connect ( toggled => \&showpercent );
$toggle3->signal_connect ( toggled => \&invert );

my $increment = 0.01;
my $timer = Glib::Timeout->add (50, \&update);

$window->show_all;
Gtk3->main;

sub showtext {
    my ($widget, $data) = @_;
    my $text = undef;
    if ( $widget->get_active ) {
        $text = "some text";
    }
    $progress->set_text( $text );
    $progress->set_show_text( $widget->get_active );
    return FALSE;
}


sub showpercent {
    my ($widget, $data) = @_;
    my $show = $widget->get_active;
    $progress->set_text( undef );
    $progress->set_show_text( $show );
    return FALSE;
}

sub invert {
    my ($widget, $data) = @_;
    $progress->set_inverted( $widget->get_active );
    return FALSE;
}

sub update {
    if ( $progress->get_fraction >= 1 ) {
        $increment = -0.01;
    }
    if ( $progress->get_fraction <= 0 ) {
        $increment = 0.01;
    }
    $progress->set_fraction( $progress->get_fraction + $increment );
    return TRUE;
}

```

8.b Spinner



```
#!/usr/bin/perl

use strict;
use warnings;
use diagnostics;
use feature ':5.14';
use Gtk3 '-init';
use Glib qw/TRUE FALSE/;

my $window = Gtk3::Window->new('toplevel');
$window->set_title("Spinner Example");
$window->set_position("mouse");
$window->set_default_size(250, 100);
$window->set_border_width(5);
$window->signal_connect (delete_event => sub { Gtk3->main_quit });

my $vbox = Gtk3::Box->new("vertical", 5);
$window->add($vbox);

my $spinner = Gtk3::Spinner->new;
my $toggle1 = Gtk3::ToggleButton->new_with_label('Spin');
$toggle1->set_active( FALSE );

$vbox->pack_start($spinner, TRUE, TRUE, 5 );
$vbox->pack_start($toggle1, TRUE, TRUE, 5 );

$toggle1->signal_connect ( toggled => \&toggle );

$window->show_all;
Gtk3->main;

sub toggle {
    my ($widget, $data) = @_;
    if ( $widget->get_active == TRUE ) {
        $spinner->start;
        $toggle1->set_label("Stop");
    } else {
        $spinner->stop;
        $toggle1->set_label("Spin");
    }
    return FALSE;
}
```

8.c **Switch**

9. **Cairo Drawing Basics**

10. **Dealing With User Input**